

Lawrence Berkeley National Laboratory

Recent Work

Title

APERTURE OF TWO-COUNTER TELESCOPES

Permalink

<https://escholarship.org/uc/item/4rr7x7v0>

Author

Swanson, W.P.

Publication Date

1988-04-01

e-2



Lawrence Berkeley Laboratory

UNIVERSITY OF CALIFORNIA

OCCUPATIONAL HEALTH DIVISION

Aperture of Two-Counter Telescopes

W.P. Swanson


April 1988

RECEIVED
LAWRENCE
BERKELEY LABORATORY
OCT 11 1988

LIBRARY AND
DOCUMENTS SECTION

TWO-WEEK LOAN COPY

*This is a Library Circulating Copy
which may be borrowed for two weeks.*



LBL-25715
e-2

DISCLAIMER

This document was prepared as an account of work sponsored by the United States Government. While this document is believed to contain correct information, neither the United States Government nor any agency thereof, nor the Regents of the University of California, nor any of their employees, makes any warranty, express or implied, or assumes any legal responsibility for the accuracy, completeness, or usefulness of any information, apparatus, product, or process disclosed, or represents that its use would not infringe privately owned rights. Reference herein to any specific commercial product, process, or service by its trade name, trademark, manufacturer, or otherwise, does not necessarily constitute or imply its endorsement, recommendation, or favoring by the United States Government or any agency thereof, or the Regents of the University of California. The views and opinions of authors expressed herein do not necessarily state or reflect those of the United States Government or any agency thereof or the Regents of the University of California.

APERTURE OF TWO-COUNTER TELESCOPES

William P. Swanson

Occupational Health Division
Lawrence Berkeley Laboratory
University of California
Berkeley, CA 94720

ABSTRACT

Calculation of the geometrical aperture of two-counter telescopes using circular counters is discussed. A FORTRAN program OVERLAP is described that performs such calculations with convenient user input. Representative results are shown and the LBL Berklet detector is calculated as a specific example. A listing of the program is given in an appendix.

April 3, 1988

APERTURE OF TWO-COUNTER TELESCOPES

William P. Swanson

Occupational Health Division
Lawrence Berkeley Laboratory
University of California
Berkeley, CA 94720

INTRODUCTION

The telescope considered consists of two coaxial circular counters (Fig. 1). The "aperture" $A_p \equiv \langle \Delta A \cdot \Delta \Omega \rangle$ is defined as the scalar quantity that satisfies the equation

$$C = A_p \times \phi_{\text{isotropic}} \quad (1)$$

where C is the number of particles that pass through both counters ("counts"), and $\phi_{\text{isotropic}}$ is a broad uniform isotropic fluence as defined by the ICRU (1977). If $\phi_{\text{isotropic}}$ is in units of $\text{cm}^{-2} \text{sr}^{-1}$, A_p will have units of $\text{cm}^2 \text{sr}$. Although the author was assured by experts that this problem had been solved many times over, a specific reference showing practical results could not be found. This note solves only the simple geometrical problem; particle "loss" by scattering or interactions in the telescope are not considered.

METHOD OF CALCULATION

"Pretend you are very far away and shining a flashlight on the smaller counter. Its shadow on the plane of the larger counter will be circular. Now move the flashlight around in all possible directions"* Program OVERLAP, written in FORTRAN for the VAX computer, evaluates the integral

$$\int \text{AREA}(\theta, \phi) d\Omega \quad (2)$$

AREA is the overlap area of the shadow on the disk of the larger counter and $d\Omega = 2\pi \sin\theta d\theta$, where θ is the polar angle to the telescope axis (Fig. 1).

For $\theta < \theta_1 \equiv \tan^{-1}[(R_2 - R_1)/S]$, $\text{AREA} = \pi(R_1)^2$, and

For $\theta > \theta_2 \equiv \tan^{-1}[(R_2 + R_1)/S]$, $\text{AREA} \equiv 0$,

(*) D. E. Groom, SSC Central Design Group, Private communication.

where R_1 and R_2 are the radii of the smaller and larger counter, respectively, and S is the counter separation. For the intermediate case, $\theta_1 < \theta < \theta_2$, the overlap area is the sum of two segments defined by the intersection of the two circles in the plane of the larger counter (Fig. 2). In this plane, the center of the smaller circle is at $(0, YC)$, where $YC = S \times \tan\theta$. The two points of intersection, (XI, YI) , are solved for by:

$$YI = (R_2^2 - R_1^2 + YC^2)/(2 \times YC). \quad (3a)$$

Then, XI is found using the equation of the larger circle:

$$XI = \pm\sqrt{(R_2^2 - YI^2)}. \quad (3b)$$

The segment area for the larger circle is found from

$$AREA_2 = R_2^2 (\phi_2 - \sin\phi_2) / 2.0 \quad (4)$$

where $\phi_2 = 2 \sin^{-1} | XI / R_2 | < \pi$ is the angle subtended by the segment of the larger circle at the circle's center $(0, 0)$. The same considerations are used to find $AREA_1$ (segment of the smaller circle) but in this case attention must be paid to the quadrant in determining ϕ_1 from the FORTRAN arcsin routine because ϕ_1 can exceed π . Then $AREA = AREA_1 + AREA_2$.

The result of the entire calculation is the summation

$$APERT = \sum (2\pi \sin\theta \Delta\theta) \times (AREA \times \cos\theta), \quad (5)$$

where APERT is the desired aperture for the counter telescope and $\cos\theta$ is the obliquity factor. As written, OVERLAP uses 100 equal intervals in θ , so that $\Delta\theta = |\theta_2 - \theta_1| / 100$.

LIMITING CASE: SEPARATION \rightarrow ZERO

For the two counters in contact ($S = 0$), we have the equivalent of a single counter of area $\pi(R_1^2)$. All particles passing through the smaller counter are registered but the relevant area will appear smaller for larger values of θ by an obliquity factor equal to $\cos\theta$. The result can be shown to be π multiplied by the area of the smaller counter. In the results shown in Figs. 3 and 4, which involve a smaller counter of area $\pi(1 \text{ cm})^2$, all curves converge to the value $\pi^2 \text{ cm}^2 \text{ sr} = 9.87 \text{ cm}^2 \text{ sr}$ as the separation goes to zero. The counters need not be circular for this limit to hold.

LIMITING CASE: SEPARATION \rightarrow LARGE

For very large separations the aperture is given by

$$A_p = A_1 A_2 / S^2 \quad (6)$$

where A_1 and A_2 are the areas of the two counters and S is the separation. Units will be

steradians times the unit of area. This is shown in Fig. 5 where the dimensionless quantity $S^2 A_p / (A_1 A_2)$ is plotted as a function of counter separation and for various ratios of R_2 / R_1 . Unity is approached for large separations. The counters need not be circular for this limit to hold.

RUNNING THE PROGRAM OVERLAP

From any VAX terminal, invoke the procedures FORTRAN, LINK and RUN using the source file OVERLAP.FOR. During the RUN procedure OVERLAP will ask for, in turn, R_1 (radius of the smaller counter), R_2 (radius of the larger counter) and S (separation of the two counters). Units are cm. A single line of answers will appear almost immediately, containing:

R1, R2, S and APERT

OVERLAP will then request a new set of input parameters and so on, *ad infinitum*. Exit from this cycle by typing CNTL C.

An example for which this program was designed is the Berklet detector (Llacer *et al.*, 1984). For this instrument, used to characterize beams of heavy ions, the parameters are $R_1 = 0.25$ cm, $R_2 = 0.55$ cm and $S = 1.0$ cm. The aperture for these parameters is calculated to be 0.138 cm² sr.

Another file, TELE.TOP, is also generated containing pairs of values (S, APERT). Each pair will appear on a separate line in a form easily adapted for input to TOPDRAWER (Chaffee, 1980). To properly invoke TOPDRAWER, some editing on TELE.TOP is required.

Examples of results from OVERLAP, plotted by TOPDRAWER are shown in Figs. 3 - 5. These may be applied to other geometries by simple scaling.

REFERENCES

- ICRU (1977). International Commission on Radiological Protection, Recommendations of the International Commission on Radiological Protection, ICRP Publication 26 (Pergamon, Oxford, 1977).
- R. B. Chaffee (1980). "Introduction to Top Drawer or A Lot of Things You Always Wanted To Know About Drawing Graphs, but Couldn't Wade Through the Top Drawer Manual to Find Out," SLAC Computation Group, Report No. CGTM-189, Stanford Linear Accelerator Center, Stanford, CA (revised Nov., 1980).
- J. Llacer, C. A. Tobias, W. R. Holley and T. Kanai (1984). "On-line characterization of heavy-ion beams with semiconductor detectors," *Med. Phys.* **11**(3), 266-278 (1984).

FIGURE CAPTIONS

- Fig. 1 Diagram defining parameters of a telescope consisting of two circular counters. Values shown are for the Berklet detector.
- Fig. 2 Diagram in plane of larger counter showing how the overlap area is calculated from the sum of two segments, one from each circle.
- Fig. 3 Output of the program OVERLAP showing behavior of aperture as a function of counter separation S and for various radii of the larger counter. For this graph, radius of the smaller counter is $R_1 = 1$ cm; other lengths are shown in units of R_1 . The limit for zero separation, π^2 cm² sr, is shown. The single point is the result for the Berklet detector.
- Fig. 4 Same as previous figure but with expanded scale.
- Fig. 5 The dimensionless quantity $A_p S^2 / (A_1 A_2)$ as a function of counter separation S and for various ratios R_2 / R_1 . For large separations the limit is 1.0.

APPENDIX: LISTING OF FORTRAN PROGRAM OVERLAP

```

C   PROGRAM OVERLAP COMPUTES THE APERTURE OF 2-COUNTER TELESCOPES
C   CONSISTING OF TWO CO-AXIAL CIRCULAR COUNTERS.

C   S IS SEPARATION OF COUNTERS; UNITS ARE CM.
C   R2, R1 ARE RADII OF LARGER, SMALLER, CIRCLE RESPECTIVELY.
C   R2 MUST BE LARGER THAN R1.

C   TO TEST THE PROGRAM, TRY THE PARAMETERS OF THE BERKLET DETECTOR:
C   R1 = 0.25, R2 = 0.55, S = 1.0 CM; RESULT: APERTURE = 0.138 CM2 SR.

DATA NTHETA / 100 /

C   THE FOLLOWING STATEMENTS CALL IN THE INPUT DATA FROM THE TERMINAL.

1   TYPE 2
2   FORMAT(//1X, 'PROGRAM OVERLAP:  COMPUTES APERTURE OF 2-COUNTER TELE
1SCOPE; '
2   / 1X '                CIRCULAR COUNTERS ARE ASSUMED.'
3   // 1X, ' TYPE RADIUS OF SMALLER COUNTER [cm] : '      )
ACCEPT*, R1
TYPE 3
3   FORMAT(/ 1X, ' TYPE RADIUS OF LARGER COUNTER [cm]'      )
ACCEPT*, R2
TYPE 4
4   FORMAT(/ 1X, ' TYPE SEPARATION BETWEEN COUNTERS [cm]'  )
ACCEPT*, S

C   THETA IS IN RADIANS
THETA   = 0.0
THETA2  = ATAN((R2 + R1) / S)
DTHETA  = THETA2 / NTHETA
APERT   = 0.0

C   PRINT 1902, R1, R2, S
1902 FORMAT(/ ' CONFIRM INPUT DATA: ' 12E10.3)
C   PRINT 1903, THETA2, DTHETA
1903 FORMAT( / ' THETA2, DTHETA = ' 2E10.3 / )
1901 FORMAT(12E10.3 /)

DO 1000 I = 1, NTHETA

THETA = THETA + DTHETA
IF (THETA . GT . THETA2) GO TO 1001

C   COORDINATES OF CENTER OF SMALLER CIRCLE AS PROJECTED ONTO SECOND PLANE:
C   S DENOTES SEPARATION BETWEEN PLANES OF CIRCLES

YC = S * TAN(THETA)

C   WE NOW HAVE TWO CIRCLES IN THE PLANE OF THE SECOND COUNTER
C   LARGER CIRCLE IS CENTERED AT (0, 0) WITH RADIUS R2.
C   SMALLER CIRCLE IS CENTERED AT (0, YC) WITH RADIUS R1.

C   FIND POINTS OF INTERSECTION
C   AXIS OF SYMMETRY IS THE Y-AXIS

```

```

AREA1 = 0.0
AREA2 = 0.0
PHI1  = 0.0
PHI2  = 0.0

```

C SOLVE FOR YI:

```
YI = (R2**2 - R1**2 + YC**2) / (2.0 * YC)
```

C SOLVE FOR XI1, XI2 USING EQUATION OF LARGER CIRCLE

```

XI = 0.0
XSQ = R2**2 - YI**2
IF (XSQ . LE . 0.0) GO TO 899

```

```
XI = SQRT(XSQ)
```

C NOW WE HAVE COORDINATES OF INTERSECTION POINTS: (-XI, YI) AND (XI, YI).

C AREA OF SEGMENT BOUNDED BY LARGER CIRCLE (2):

```

IF (YI . LE . 0.000001) GO TO 899
PHI2 = 2.0 * ASIN(XI / R2)

AREA2 = R2 * R2 * (PHI2 - SIN(PHI2)) / 2.0

```

C AREA OF SEGMENT BOUNDED BY SMALLER CIRCLE (1):

```

AMY = YC - YI
IF (ABS (AMY) . LE . 0.000001) GO TO 899

```

C PHI1 FROM FORTRAN SUBROUTINE ASIN WILL ALWAYS BE POSITIVE.

```
PHI1 = 2.0 * ASIN(XI / R1)
```

C PUT PHI1 INTO CORRECT QUADRANT.

```
IF (AMY . LE . 0.0) PHI1 = 2.0 * 3.1415927 - PHI1
```

```
AREA1 = R1 * R1 * (PHI1 - SIN(PHI1)) / 2.0
```

C NOW FIND TOTAL AREA:

```
AREA = AREA1 + AREA2
```

```

899 IF (YC + R1 . LE . R2) AREA = 3.14159 * R1 * R1
IF (YC - R1 . GE . R2) AREA = 0.0

```

C USE AREA OF SMALLER CIRCLE AS SEEN BY RAYS INCIDENT AT ANGLE THETA.

```
999 IF (AREA . LE . 0.0) GO TO 1001
```

```

DWTAREA = 2.0 * 3.1415927 * SIN(THETA) * DTHETA * AREA
APERT    = APERT + DWTAREA * COS(THETA)

```

C THE FOLLOWING DE-ACTIVATED STATEMENTS WERE USED FOR DEBUGGING.

```

C PRINT 1901, THETA, YC, YI, XI, PHI1, PHI2,
C 1 AREA1, AREA2, AREA, DWTAREA, APERT

```

```
1000 CONTINUE
1001 CONTINUE
```

```
C INPUT PARAMETERS AND RESULT ARE PRINTED ON THE TERMINAL SCREEN.
```

```
PRINT 1905, R1, R2, S, APERT
```

```
1905 FORMAT(' R1, R2, S =' 3E10.3,
1 ' APERT =' E12.4, ' [sr cm2]' // )
```

```
C A FILE IS WRITTEN THAT CAN BE USED FOR TOPDRAWER INPUT.
```


```
OPEN (UNIT=11, NAME='TELE.TOP', ERR=1, TYPE='NEW')
```

```
WRITE(11, 1906) S, APERT
```

```
1906 FORMAT(2E12.4)
```


```
GO TO 1
```

```
END
```




XBL 884-9602

Fig. 1 Diagram defining parameters of a telescope consisting of two circular counters. Values shown are for the Berklet detector.




XBL 884-9603

Fig. 2 Diagram in plane of larger counter showing how the overlap area is calculated from the sum of two segments, one from each circle.



XBL 884-9615


Fig. 3 Output of the program OVERLAP showing behavior of aperture as a function of counter separation S and for various radii of the larger counter. For this graph, radius of the smaller counter is $R_1 = 1$ cm; other lengths are shown in units of R_1 . The limit for zero separation, π^2 cm² sr, is shown. The single point is the result for the Berklet detector.



XBL 884-9614

Fig. 4 Same as previous figure but with expanded scale.

Two-counter telescope



XBL 884-9613

Fig. 5 The dimensionless quantity $A_p S^2/(A_1 A_2)$ as a function of counter separation S and for various ratios R_2/R_1 . For large separations the limit is 1.0.

*LAWRENCE BERKELEY LABORATORY
TECHNICAL INFORMATION DEPARTMENT
UNIVERSITY OF CALIFORNIA
BERKELEY, CALIFORNIA 94720*